



Formation SIN

Utilisation du logiciel Basic4Android

Formateur : Pierre Aguillon



PARTIE 4 : PROGRAMME 4

Objectifs :

- Mettre en œuvre une communication en TCP Client ;
- Utiliser le logiciel gratuit Hercules ;
- Découvrir la librairie RandomAccessFile;

Programme 4 :

Ce quatrième programme va nous permettre de rendre communicante notre tablette via le réseau Wifi en utilisant le protocole TCP.

Pour pouvoir mettre en œuvre les programmes, il nous faut l'environnement suivant :

- Une tablette connectée en Wifi sur un réseau et dont on connaît l'adresse IP attribuée par le serveur DHCP ;
- Un ordinateur connecté (peu importe le type de connexion) sur le même réseau et dont on connaît l'adresse IP attribuée par le serveur DHCP (IPCONFIG) ;
- Le logiciel Hercules de chez HWgroup que l'on peut télécharger ici http://www.hw-group.com/products/hercules/index_en.html ;

En TCP le principe est plus complexe qu'UDP. Il faut d'abord définir si votre tablette sera client ou serveur TCP. Généralement, dans nos applications, elle est « Client ». On va donc limiter l'étude à cette fonctionnalité.

Pour cela il faut se remémorer le principe de la connexion en TCP :

- Le serveur crée un port TCP (plus exactement un socket) puis il le met en écoute – ici c'est le logiciel Hercules qui va s'en charger ;
- Le client fait une demande de connexion ;
- Le serveur valide cette demande ;
- L'échange des données est effectif ;
- Le client fait une demande de fermeture du socket ;
- Le serveur ferme le socket et le repositionne en écoute pour une future connexion ;

Dans les exercices, il faudra penser à bien changer les adresses IP en positionnant celles correspondant à votre configuration. Elles seront surlignées en rouge.

Créez un nouveau programme que vous appellerez « programme n4 ».

L'utilisation du protocole TCP nécessite l'utilisation des bibliothèques **NetWork** et **RandomAccessFile**. Cochez-les dans votre liste de librairie.

Lancez le Designer.

On va construire une feuille contenant :

- 1 bouton : Name=send ; Text=Send
- 1 bouton : Name=connect ; Text=Connexion
- 1 bouton : Name=deconnect ; Text=Déconnexion
- 1 label : Name=label1 ;
- 1 EditText : Name=edittext1 ;

Placez de la manière suivante :



Générez les membres, seuls les **click** sur les trois boutons nous sont utiles. Sauvegardez votre feuille sous le nom **layout1** et fermez le Designer.

Les fonctionnalités de notre programme devront être les suivantes :

- Le texte du **label1** prendra la valeur « connecté » si le client est connecté et « non connecté » si le client ne l'est pas ;
- Le bouton **connect** doit permettre la connexion ;
- Le bouton **deconnect** la déconnexion ;
- Le bouton **send** l'envoi de deux octets ;
- Lorsque le client est connecté seuls les boutons **deconnect** et **send** doivent être visibles ;
- Lorsque le client est déconnecté seul le bouton **connect** doit être visible.

En premier lieu on va placer les lignes de codes exécutées lors du lancement de l'application sous Android. Dans ce cas la connexion n'est pas réalisée, ce qui explique les lignes de codes encadrées en rouge (*note : on aurait pu les placer dans le Designer*)

```

23 Sub Activity_Create(FirstTime As Boolean)
24     Activity.LoadLayout("layout1")
25     Label1.Text="Non connecté"
26     Connect.Visible=True
27     deconnect.Visible=False
28     Send.Visible=False
29     EditText1.Visible=False
30 End Sub

```

L'utilisation en client du protocole TCP sous B4A se passe en cinq étapes : la création du socket, son initialisation (adresse IP, port, la taille mémoire allouée au buffer), la connexion du socket (en TCP on travaille en mode connecté), l'échange des données (si le socket est connecté), et enfin la fermeture du socket.

La création du socket : on définit la variable **Mon_socket** comme étant un **Socket** (contrairement à l'UDP où l'on doit préciser **UDPSocket**, le type **Socket** seul correspond au TCP). Lorsque l'on utilise un Socket, on dispose de deux propriétés importantes :

- Socket.inputStream : c'est le flux de données entrantes dans le socket (la réception de données)
- Socket.outputStream : c'est le flux sortant (l'émission de données) ;

```

9 Sub Globals
10 'These global variables will be redeclared each time the activity is created.
11 'These variables can only be accessed from this module.
12 Dim Mon_Socket As Socket
13 Dim flux As AsyncStreams
14 Dim Send As Button
15 Dim Labell As Label
16 Dim Connect As Button
17 Dim deconnect As Button
18 Dim EditText1 As EditText
19 Dim bc As ByteConverter
20 Dim Label2 As Label
21 End Sub

```

L'initialisation du socket et la tentative de connexion : C'est l'appui sur le bouton Connect qui doit initialiser le Socket et faire une tentative de connexion sur le serveur dont on connaît l'adresse IP (ici celle de votre PC)

```

59 Sub Connect_Click
60 Mon_Socket.Initialize("Mon_Socket")
61 Mon_Socket.Connect("192.168.1.90", 5500, 500)
62 End Sub

```

S'il y a une tentative de connexion, un événement se déclenche : il s'agit de la fonction **Mon_Socket_Connected(Succes As Boolean)**. Celle-ci renvoie une variable (**Succes**) qui peut prendre deux valeurs **True** ou **False**, si la connexion est effective ou pas. Suivant ce résultat on va modifier la propriété de nos éléments pour répondre aux exigences du programme.

```

40 Sub Mon_Socket_Connected(Succes As Boolean)
41 If Succes=True Then
42 Labell.text="Connecté"
43 Connect.Visible=False
44 deconnect.Visible=True
45 Send.Visible=True
46 EditText1.Visible=True
47 flux.Initialize(Mon_Socket.InputStream, Mon_Socket.OutputStream, "flux")
48 Else
49 Labell.text="Non connecté"
50 Connect.Visible=True
51 deconnect.Visible=False
52 Send.Visible=False
53 EditText1.Visible=False
54 End If
55 End Sub

```

L'échange de données : c'est ici que la librairie **RandomAccessFile** va nous être utile.

Cette dernière propose un objet, nommé **AsyncStreams**. **AsyncStreams** vous permet de lire des données d'un **InputStream** et d'écrire des données dans un **OutputStream**, sans bloquer votre programme.

Quand de nouvelles données sont disponibles sur l'**InputStream**, l'événement **NewData** est déclenché avec les données.

Quand vous écrivez des données à l'**OutputStream**, elles sont ajoutées à une file d'attente interne et envoyées ensuite en arrière-plan.

AsyncStreams est très utile pour des flux de données lents tels que les réseaux Ethernet ou Bluetooth.

Pour utiliser l'objet **AsyncStreams**, il faut d'abord le définir (1) puis l'associer au flux de données de notre socket lorsque la connexion est effective (2)

```

9 Sub Globals
10     'These global variables will be redeclared each time the activity is created.
11     'These variables can only be accessed from this module.
12     Dim Mon_Socket As Socket
13     Dim flux As AsyncStreams 1
14     Dim Send As Button
15     Dim Labell1 As Label
16     Dim Connect As Button
17     Dim deconnect As Button
18     Dim EditText1 As EditText
19     Dim bc As ByteConverter
20     Dim Labell2 As Label
21 End Sub

```

```

40 Sub Mon_Socket_Connected(Succes As Boolean)
41     If Succes=True Then
42         Labell1.text="Connecté"
43         Connect.Visible=False
44         deconnect.Visible=True
45         Send.Visible=True
46         EditText1.Visible=True
47         flux.Initialize(Mon_Socket.InputStream, Mon_Socket.OutputStream, "flux") 2
48     Else
49         Labell1.text="Non connecté"
50         Connect.Visible=True
51         deconnect.Visible=False
52         Send.Visible=False
53         EditText1.Visible=False
54     End If
55 End Sub

```

Pour l'émission : si le bouton **Send** est visible, c'est qu'une connexion est valide. C'est donc dans l'évènement **Send_Click** que l'on va réaliser l'émission de notre message (ici deux octets).

On va d'abord créer le tableau des données à transmettre (1) ; puis on écrit ces valeurs dans ce flux (2).

```

90 Sub Send_Click
91     Dim emission(2) As Byte
92     emission(1)=0x34 1
93     emission(0)=0x35
94
95     flux.Write(emission) 2
96
97
98 End Sub

```

*Note : on aurait pu vérifier au début de la fonction si la connexion était bien valide en testant si **Mon_Socket.Connected=True**.*

Pour la réception : si des données nouvelles arrivent sur le **InputStream** associé à notre objet **flux**, un événement est déclenché et la fonction **flux_newdata(buffer() as Byte)** est exécutée.

Les données entrantes sont stockées dans le tableau d'octets défini lors du déclenchement

de la fonction : **buffer()**. Pour pouvoir afficher ces données en Ascii dans la propriété **label2.text**, il faut les convertir en chaîne de caractères. C'est le rôle de la fonction **BytesToString** où l'on doit préciser : le nom du tableau d'octets (**buffer**) ; à partir de quelle position on commence la conversion (**offset**, ici 0) ; jusqu'à quelle position on fait la conversion (tout le tableau, **buffer.length**) ; le format de conversion (« **ASCII** »).

Le programme donne donc :

```

33
34 Sub flux_newdata(buffer() As Byte)
35     Dim msg As String
36     msg=BytesToString(buffer,0,buffer.length,"ascii")
37     Label2.Text=msg
38 End Sub
39

```

Reste encore à gérer l'appui sur le bouton poussoir **deconnect** devant assurer la déconnexion du socket et la remise en place des propriétés des objets :

```

74 Sub deconnect_Click
75     Mon_Socket.Close
76     Label1.text="Non connecté"
77     Connect.Visible=True
78     deconnect.Visible=False
79     Send.Visible=False
80     EditText1.Visible=False
81 End Sub
82

```

Compilez votre programme et exécutez.

Sur votre PC, lancez Hercules et sous l'onglet TCP_SERVER saisissez le numéro du port 5500. Votre tablette doit pouvoir échanger des données avec le PC.

On vient de voir comment envoyer des octets. Si l'on veut envoyer du texte que l'utilisateur pourra saisir dans l'objet **EditText**, il va falloir modifier le programme pour réaliser des manipulations de variables.

```

57 Sub Send_Click
58     Dim emission() As Byte
59     Dim bc As ByteConverter
60     Dim message As String
61     message=EditText1.text
62     emission=bc.StringToBytes(message,"ascii")
63
64     flux.Write(emission)
65
66 End Sub
67

```

*Note : il faut que la bibliothèque optionnelle **ByteConverter** soit sélectionnée (voir programme 3 en UDP)*